// File: contracts/MeaningOfLife.sol



/*
      \  |                     _)                     _ \    _|      |     _)   _|                    
     |\/ |   _ \   _` |  __ \   |  __ \    _` |      |   |  |        |      |  |     _ \              
     |   |   __/  (   |  |   |  |  |   |  (   |      |   |  __|      |      |  __|   __/              
    _|  _| \___| \__,_| _|  _| _| _|  _| \__, |     \___/  _|       _____| _| _|   \___|              
                                         |___/
   ___|         |  |                       |       ___|                |                      |       
  |      |   |  |  __|  |   |   __|  _` |  |      |       _ \   __ \   __|   __|  _` |   __|  __|     
  |      |   |  |  |    |   |  |    (   |  |      |      (   |  |   |  |    |    (   |  (     |       
 \____| \__,_| _| \__| \__,_| _|   \__,_| _|     \____| \___/  _|  _| \__| _|   \__,_| \___| \__|     

by maciej wisniewski                                                                                                                   
*/

// SPDX-License-Identifier: MIT

pragma solidity 0.8.11;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC721/extensions/ERC721Enumerable.sol";
import "@openzeppelin/contracts/utils/Strings.sol";
import "base64-sol/base64.sol";

/**
 * @dev Required interface of a CulturalContract compliant contract.
 */
interface ICulturalContract {
    /**
     * @dev Returns cRNA number, cRNA length and cRNA unit length.
     *
     * cRNA defines the attributes of the ERC721 token.
     * cRNA is divided in equal length units. Every unit defines a set of attributes.
     * cRNAs can be easily manipulated using simple mathematical operations providing
     * many creative possibilities for remixes and mashups.
     *
     */
    function culturalRNA(uint256 tokenId)
        external
        returns (
            uint256 cRNA,
            uint256 numberLength,
            uint256 unitLength
        );

    /**
     * @dev Remixes the ERC721 token based on the `input`.
     */
    function remix(uint256 tokenId, string memory input) external;

    /**
     * @dev Returns mashup result of ERC721 token from a given `mashupAddress`.
     */
    function mashup(uint256 tokenId, address payable mashupAddress)
        external
        payable
        returns (string memory);

    /**
     * @dev Returns `mashupFee` from `mashupAddress`.
     */
    function mashupFee(address mashupAddress) external returns (uint256);
}

/**
 * @dev Optional interface of a CulturalContract compliant contract.
 */
interface ICulturalContractMetadata {
    /**
     * @dev Returns the CulturalContract name.
     */
    function ccName() external view returns (string memory);

    /**
     * @dev Returns the CulturalContract symbol.
     */
    function ccSymbol() external view returns (string memory);

    /**
     * @dev Returns the CulturalContract author.
     */
    function ccAuthor() external view returns (string memory);
}

abstract contract CulturalContract is
    ICulturalContract,
    ICulturalContractMetadata,
    ERC721Enumerable,
    ReentrancyGuard,
    Ownable
{
    event MashedUp(uint256 indexed tokenId);
    event Remixed(uint256 indexed tokenId);

    struct CulturalRNA_Unit {
        uint256 numberLength;
        uint256 unitLength;
    }

    /**
     * @dev Helper function. Returns cRNA unit array for a given cRNA number.
     */
    function decode(
        uint256 number,
        uint256 numberLength,
        uint256 unitLength
    ) internal pure returns (uint256[] memory) {
        uint256[] memory units = new uint256[](numberLength / unitLength);
        uint256 i;
        uint256 counter = 0;
        for (i = 0; i < numberLength / unitLength; i++) {
            units[i] =
                ((number % (10**(numberLength - counter))) /
                    (10**(numberLength - (counter + unitLength)))) %
                10**unitLength;
            counter = counter + unitLength;
        }
        return units;
    }

    /**
     * @dev Helper function. Returns cRNA number for a given cRNA unit array.
     */
    function encode(uint256[] memory numbers, uint256 unitLength)
        internal
        pure
        returns (uint256)
    {
        uint256 encodedNumber = numbers[0];
        uint256 i;
        for (i = 0; i < numbers.length - 1; i++) {
            encodedNumber = encodedNumber * 10**unitLength + numbers[i + 1];
        }
        return encodedNumber;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     *
     * Note: the ERC-165 identifier for ICulturalContract interface is 0xf9ae973e
     */
    function supportsInterface(bytes4 interfaceId)
        public
        view
        virtual
        override
        returns (bool)
    {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            interfaceId == type(ICulturalContract).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev Helper function. Returns ICulturalContract interface identifier.
     */
    function getICCinterface() public pure returns (bytes4) {
        return type(ICulturalContract).interfaceId;
    }
}

contract MeaningOfLife is CulturalContract {
    event SalePaused(bool paused);

    string public constant DESCRIPTION =
        "Meaning-of-Life is an artwork created by Maciej Wisniewski in 2002 for the Whitney Museum of American Art and made available by the artist as a CulturalContract in 2022. CulturalContracts are NFTs that can be remixed and mashed up.";

    CulturalRNA_Unit public cRNA_U =
        CulturalRNA_Unit({numberLength: 35, unitLength: 5});

    mapping(uint256 => uint256) private tokenCRNA;
    mapping(string => uint256) private propertyIndex;
    mapping(uint256 => bool) private isRemixed;
    mapping(uint256 => string) private remixValue;
    mapping(uint256 => bool) private isMashedUp;
    mapping(uint256 => string) private mashupValue;
    mapping(uint256 => address) private mashedUpBy;
    mapping(address => bool) private inMashupRegistry;
    mapping(address => uint256) private mashupRanking;
    mapping(address => uint256) private mashupPrice;
    mapping(address => string) private mashupName;
    mapping(address => string) private mashupSymbol;
    mapping(address => string) private mashupAuthor;
    mapping(address => uint256) private allMashupsIndex;
    mapping(address => uint256) private latestMashupTokenId;

    uint256 public constant MAX_TOKENS = 12000;
    uint256 public constant MAX_RESERVED_TOKENS = 480;
    uint256 public price = 100000000000000000; //0.1 ETH

    address[] private allMashups;
    uint256 private numRemixMashups = 0;
    bool private paused = false;
    string private author = "maciej wisniewski";
    uint256 private tokenCounter = 0;
    uint256 private reservedTokenCounter = 0;

    string[] private locations = [
        "Acapulco",
        "Aklavik",
        "Algiers",
        "Almaty",
        "Anadyr",
        "Anchorage",
        "Yerevan",
        "Asuncion",
        "Atlanta",
        "Bandung",
        "Bangkok",
        "Beijing",
        "Bogota",
        "Boston",
        "Brasilia",
        "Brisbane",
        "Buenos Aires",
        "Canberra",
        "Caracas",
        "Casablanca",
        "Chicago",
        "Chongquing",
        "Denpasar",
        "Denver",
        "Detroit",
        "Dublin",
        "Edmonton",
        "Pyongyang",
        "Guatemala City",
        "Hangzhou",
        "Hanoi",
        "Harare",
        "Harrisburg",
        "Havana",
        "Hong Kong",
        "Honiara",
        "Honolulu",
        "Houston",
        "Indianapolis",
        "Iqaluit",
        "Istanbul",
        "Jakarta",
        "Kamchatka",
        "Kingston",
        "Kingstown",
        "Kiritimati",
        "Kolonia",
        "Kuala Lumpur",
        "Kuwait City",
        "La Paz",
        "Lagos",
        "Lima",
        "Lisbon",
        "London",
        "Los Angeles",
        "Malang",
        "Managua",
        "Manila",
        "Mataram",
        "Medan",
        "Melbourne",
        "Mexicali",
        "Mexico City",
        "Minneapolis",
        "Montgomery",
        "Montreal",
        "Moscow",
        "Nagoya",
        "Nairobi",
        "Nassau",
        "New Orleans",
        "New York City",
        "Noumea",
        "Novosibirsk",
        "Odessa",
        "Omsk",
        "Ottawa",
        "Palembang",
        "Perth",
        "Philadelphia",
        "Phnom Penh",
        "Phoenix",
        "Prague",
        "Reykjavik",
        "Rio de Janeiro",
        "Chiang Mai",
        "San Francisco",
        "San Juan",
        "San Salvador",
        "Santiago",
        "Santo Domingo",
        "Sao Paulo",
        "Seattle",
        "Semarang",
        "Seoul",
        "Shanghai",
        "Singapore",
        "St. Paul",
        "Stockholm",
        "Surabaya",
        "Surakarta",
        "Suva",
        "Sydney",
        "Taipei",
        "Tangshan",
        "Tegucigalpa",
        "The Settlement",
        "Tianjin",
        "Tijuana",
        "Tokyo",
        "Toronto",
        "Qingdao",
        "Ulaanbaatar",
        "Vancouver",
        "Vientiane",
        "Vladivostok",
        "Warsaw",
        "Washington DC",
        "Wellington",
        "Winnipeg",
        "Aden",
        "Aleppo",
        "Alexandria",
        "Amman",
        "Ankara",
        "Athens",
        "Baghdad",
        "Beirut",
        "Belgrade",
        "Bethlehem",
        "Bruges",
        "Cairo",
        "Cape Town",
        "Colombo",
        "Damascus",
        "Delhi",
        "Fez",
        "Florence",
        "Hyderabad",
        "Jaffa",
        "Jaffna",
        "Jerusalem",
        "Kandahar",
        "Kathmandu",
        "Kotor",
        "Lahore",
        "Lesbos",
        "Lhasa",
        "Ljubljana",
        "Madurai",
        "Mecca",
        "Minsk",
        "Mombasa",
        "Monte Carlo",
        "Mumbai",
        "Muscat",
        "Nicosia",
        "Peshawar",
        "Plovdiv",
        "Quito",
        "Riga",
        "Riyadh",
        "Sarajevo",
        "Sofia",
        "Split",
        "Srinagar",
        "Tallinn",
        "Tashkent",
        "Tbilisi",
        "Teheran",
        "Thessaloniki",
        "Timbuktu",
        "Tirana",
        "Tripoli",
        "Varanasi",
        "Venice",
        "Vienna",
        "Vilnius",
        "Yangon",
        "Zurich",
        "Agartha",
        "Amaravati",
        "Amaurotum",
        "Arcadia",
        "Arkham",
        "Asgard",
        "Atlantis",
        "Ayodhya",
        "Aztlan",
        "Brahmaloka",
        "Camelot",
        "Casterbridge",
        "Diyu",
        "Duat",
        "El Dorado",
        "Elsinore",
        "Emerald City",
        "Erewhon",
        "Gorias",
        "Jotunheim",
        "Kunlun",
        "Lorbrulgrud",
        "Mayberry",
        "Mildendo",
        "Pandaemonium",
        "Riverdale",
        "Shambala",
        "Shangri La",
        "Springfield",
        "Takamagahara",
        "Elysium",
        "Themiscyra",
        "Valhalla",
        "Xanadu",
        "Z",
        "Oslo",
        "Copenhagen",
        "Helsinki",
        "Tel Aviv",
        "Milanowek"
    ];

    string[] private timeOfDayColors = [
        "whitesmoke",
        "lightgray",
        "darkgray",
        "dimgray",
        "darkgray",
        "lightgray"
    ];
    string[] private timeOfDay = [
        "During the day of ",
        "In the early evening on ",
        "On the evening of ",
        "During the night of ",
        "In the early morning on ",
        "On the morning of "
    ];

    string[] private timeOfDayMeta = [
        "Day",
        "Early Evening",
        "Evening",
        "Night",
        "Early Morning",
        "Morning"
    ];

    string[] private months = [
        "January",
        "February",
        "March",
        "April",
        "May",
        "June",
        "July",
        "August",
        "September",
        "October",
        "November",
        "December"
    ];

    constructor() ERC721("MeaningOfLife CC", "MLCC") Ownable() {
        propertyIndex["time"] = 0;
        propertyIndex["year"] = 1;
        propertyIndex["month"] = 2;
        propertyIndex["day"] = 3;
        propertyIndex["location"] = 4;
        propertyIndex["remix"] = 5;
        propertyIndex["mashup"] = 6;
    }

    /**
     * @dev Returns a pseudorandom number.
     */
    function mix(uint256 numberLength, uint256 tokenId)
        internal
        view
        returns (uint256)
    {
        uint256 mixedNumber = uint256(
            keccak256(abi.encodePacked(block.timestamp, msg.sender, tokenId))
        ) % 10**numberLength;
        return mixedNumber;
    }

    /**
     * @dev Returns true if the given year is a leap year.
     *
     * Helper calendar function.
     */
    function isLeapYear(uint256 year) internal pure returns (uint256) {
        if (year == 1900) {
            return 1;
        } else {
            return year % 4;
        }
    }

    /**
     * @dev Returns the number of days in a given month.
     *
     * Helper calender function.
     */
    function numberOfDays(uint256 month, uint256 year)
        internal
        pure
        returns (uint256)
    {
        if (month == 3 || month == 5 || month == 8 || month == 10) {
            return 30;
        } else if (month == 1) {
            if (isLeapYear(year) == 0) {
                return 29;
            } else {
                return 28;
            }
        } else {
            return 31;
        }
    }

    /**
     * @dev Returns the unit number for a given attribute set.
     */
    function getCode(uint256 tokenId, uint256 index)
        internal
        view
        returns (uint256)
    {
        uint256 cRNAnumber = tokenCRNA[tokenId];
        uint256 arraySize = cRNA_U.numberLength / cRNA_U.unitLength;
        uint256[] memory codeArray = new uint256[](arraySize);
        codeArray = decode(cRNAnumber, cRNA_U.numberLength, cRNA_U.unitLength);
        return codeArray[index];
    }

    /**
     * @dev Adds the remix value to the cRNA.
     */
    function getRemixCRNA(uint256 tokenId, uint256 remixCRNA)
        internal
        view
        returns (uint256)
    {
        uint256 cRNAnumber = tokenCRNA[tokenId];
        uint256 arraySize = cRNA_U.numberLength / cRNA_U.unitLength;
        uint256[] memory codeArray = new uint256[](arraySize);
        codeArray = decode(cRNAnumber, cRNA_U.numberLength, cRNA_U.unitLength);
        codeArray[propertyIndex["remix"]] = remixCRNA;
        return encode(codeArray, cRNA_U.unitLength);
    }

    /**
     * @dev Adds the mashup value to the cRNA.
     */
    function getMashupCRNA(uint256 tokenId, uint256 mashupCRNA)
        internal
        view
        returns (uint256)
    {
        uint256 cRNAnumber = tokenCRNA[tokenId];
        uint256 arraySize = cRNA_U.numberLength / cRNA_U.unitLength;
        uint256[] memory codeArray = new uint256[](arraySize);
        codeArray = decode(cRNAnumber, cRNA_U.numberLength, cRNA_U.unitLength);
        codeArray[propertyIndex["mashup"]] = mashupCRNA;
        return encode(codeArray, cRNA_U.unitLength);
    }

    /**
     * @dev Returns the time of the day.
     *
     * @dev Helper calender function.
     */
    function getTimeOfDay(uint256 tokenId)
        internal
        view
        returns (string memory)
    {
        uint256 time = getCode(tokenId, propertyIndex["time"]) %
            timeOfDay.length;
        return string(abi.encodePacked(timeOfDay[time]));
    }

    /**
     * @dev Returns the time of the day meta data.
     */
    function getTimeOfDayMeta(uint256 tokenId)
        internal
        view
        returns (string memory)
    {
        uint256 time = getCode(tokenId, propertyIndex["time"]) %
            timeOfDayMeta.length;
        return string(abi.encodePacked(timeOfDayMeta[time]));
    }

    /**
     * @dev Returns the background color based on the time of the day.
     */
    function getColor(uint256 tokenId) internal view returns (string memory) {
        uint256 time = getCode(tokenId, propertyIndex["time"]) %
            timeOfDayColors.length;
        return string(abi.encodePacked(timeOfDayColors[time]));
    }

    /**
     * @dev Returns the year.
     */
    function getYear(uint256 tokenId) internal view returns (uint256) {
        return (getCode(tokenId, propertyIndex["year"]) % 150) + 1900;
    }

    /**
     * @dev Returns the month.
     */
    function getMonth(uint256 tokenId) internal view returns (string memory) {
        uint256 month = getCode(tokenId, propertyIndex["month"]) %
            months.length;
        return string(abi.encodePacked(months[month]));
    }

    /**
     * @dev Returns the location.
     */
    function getLocation(uint256 tokenId)
        internal
        view
        returns (string memory)
    {
        uint256 location = getCode(tokenId, propertyIndex["location"]) %
            locations.length;
        return string(abi.encodePacked(locations[location]));
    }

    /**
     * @dev Returns the month index.
     */
    function getMonthIndex(uint256 tokenId) internal view returns (uint256) {
        return getCode(tokenId, propertyIndex["month"]) % months.length;
    }

    /**
     * @dev Returns a day for a given month and year.
     */
    function getDay(uint256 tokenId) internal view returns (uint256) {
        uint256 year = getYear(tokenId);
        uint256 month = getMonthIndex(tokenId);
        uint256 numberDays = numberOfDays(month, year);
        return (getCode(tokenId, propertyIndex["day"]) % numberDays) + 1;
    }

    /**
     * @dev Returns the cRNA number.
     */
    function getTokenCRNA(uint256 tokenId) internal view returns (uint256) {
        return tokenCRNA[tokenId];
    }

    /**
     * @dev Returns the text for a given token including the remix if available.
     */
    function getText(uint256 tokenId) internal view returns (string memory) {
        string[11] memory textString;
        textString[0] = getTimeOfDay(tokenId);
        textString[1] = " ";
        textString[2] = getMonth(tokenId);
        textString[3] = " ";
        textString[4] = Strings.toString(getDay(tokenId));
        textString[5] = ", ";
        textString[6] = Strings.toString(getYear(tokenId));
        textString[7] = " in ";
        textString[8] = getLocation(tokenId);
        textString[9] = getRemixed(tokenId);

        string memory textCombined = string(
            abi.encodePacked(
                textString[0],
                textString[1],
                textString[2],
                textString[3],
                textString[4],
                textString[5]
            )
        );
        textCombined = string(
            abi.encodePacked(
                textCombined,
                textString[6],
                textString[7],
                textString[8],
                textString[9]
            )
        );
        return textCombined;
    }

    /**
     * @dev Returns the title of a given token.
     */
    function getName(uint256 tokenId) internal view returns (string memory) {
        string[11] memory textString;
        textString[0] = getMonth(tokenId);
        textString[1] = " ";
        textString[2] = Strings.toString(getDay(tokenId));
        textString[3] = ", ";
        textString[4] = Strings.toString(getYear(tokenId));
        string memory textCombined = string(
            abi.encodePacked(
                textString[0],
                textString[1],
                textString[2],
                textString[3],
                textString[4]
            )
        );
        return textCombined;
    }

    /**
     * @dev Returns the remix value.
     */
    function getRemixed(uint256 tokenId) internal view returns (string memory) {
        string memory remixedString;

        if (isRemixed[tokenId]) {
            uint256 remixIndex = getCode(tokenId, propertyIndex["remix"]);
            remixedString = string(
                abi.encodePacked(
                    '<tspan x="5" y="330">',
                    remixValue[remixIndex],
                    "</tspan>"
                )
            );
        } else {
            remixedString = string(abi.encodePacked("... "));
        }

        return remixedString;
    }

    /**
     * @dev Returns the mashup value.
     */
    function getMashedUp(uint256 tokenId)
        internal
        view
        returns (string memory)
    {
        string memory mashupString;

        if (isMashedUp[tokenId]) {
            uint256 mashupIndex = getCode(tokenId, propertyIndex["mashup"]);
            mashupString = string(abi.encodePacked(mashupValue[mashupIndex]));
        } else {
            mashupString = string(abi.encodePacked(""));
        }

        return mashupString;
    }

    /**
     * @dev Returns token metadata.
     */
    function getMetaData(uint256 tokenId)
        internal
        view
        returns (string memory)
    {
        string[15] memory metaInfo;
        metaInfo[0] = '[{"trait_type": "Remix", "value": "';
        if (isRemixed[tokenId]) {
            metaInfo[1] = "Yes";
        } else {
            metaInfo[1] = "No";
        }
        metaInfo[2] = '"},{"trait_type": "Mashup", "value": "';
        if (isMashedUp[tokenId]) {
            metaInfo[3] = mashupSymbol[mashedUpBy[tokenId]];
        } else {
            metaInfo[3] = "No";
        }
        metaInfo[4] = '"},{"trait_type": "Time Of Day", "value": "';
        metaInfo[5] = getTimeOfDayMeta(tokenId);
        metaInfo[6] = '"},{"trait_type": "Month", "value": "';
        metaInfo[7] = getMonth(tokenId);
        metaInfo[8] = '"},{"trait_type": "Day", "value": "';
        metaInfo[9] = Strings.toString(getDay(tokenId));
        metaInfo[10] = '"},{"trait_type": "Year", "value": "';
        metaInfo[11] = Strings.toString(getYear(tokenId));
        metaInfo[12] = '"},{"trait_type": "Location", "value": "';
        metaInfo[13] = getLocation(tokenId);
        metaInfo[14] = '"}]';

        string memory metaCombined = string(
            abi.encodePacked(
                metaInfo[0],
                metaInfo[1],
                metaInfo[2],
                metaInfo[3],
                metaInfo[4],
                metaInfo[5],
                metaInfo[6],
                metaInfo[7]
            )
        );
        metaCombined = string(
            abi.encodePacked(
                metaCombined,
                metaInfo[8],
                metaInfo[9],
                metaInfo[10],
                metaInfo[11],
                metaInfo[12],
                metaInfo[13],
                metaInfo[14]
            )
        );

        return metaCombined;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId)
        public
        view
        override
        returns (string memory)
    {
        require(
            _exists(tokenId),
            "ERC721Metadata: URI query for nonexistent token"
        );
        string[7] memory svgString;

        svgString[
            0
        ] = '<svg xmlns="http://www.w3.org/2000/svg" width="100%" height="100%" preserveAspectRatio="xMinYMin meet" viewBox="0 0 350 350"><style type="text/css">.txt{fill: black;font-family: "Open Sans", sans-serif; font-size: 0.75em;}</style><rect width="100%" height="100%" fill="';
        svgString[1] = getColor(tokenId);
        svgString[2] = '"/>';
        svgString[3] = getMashedUp(tokenId);
        svgString[4] = '<text x="5" y="310" class="txt">';
        svgString[5] = getText(tokenId);
        svgString[6] = "</text></svg>";

        string memory svgFile = string(
            abi.encodePacked(
                svgString[0],
                svgString[1],
                svgString[2],
                svgString[3],
                svgString[4],
                svgString[5],
                svgString[6]
            )
        );

        string memory jsonFile = Base64.encode(
            bytes(
                string(
                    abi.encodePacked(
                        '{"name": "',
                        getName(tokenId),
                        '", "description": "',
                        DESCRIPTION,
                        '", "attributes": ',
                        getMetaData(tokenId),
                        ', "image": "data:image/svg+xml;base64,',
                        Base64.encode(bytes(svgFile)),
                        '"}'
                    )
                )
            )
        );
        return
            string(abi.encodePacked("data:application/json;base64,", jsonFile));
    }

    /**
     * @dev See {ICulturalContract-culturalRNA}.
     */
    function culturalRNA(uint256 tokenId)
        public
        view
        override
        returns (
            uint256,
            uint256,
            uint256
        )
    {
        return (tokenCRNA[tokenId], cRNA_U.numberLength, cRNA_U.unitLength);
    }

    /**
     * @dev Pauses/Unpauses sale.
     *
     * Emits a {SalePaused} event.
     */
    function setSalePaused(bool isPaused) external onlyOwner {
        paused = isPaused;
        emit SalePaused(paused);
    }

    function getSalePaused() external view returns (bool) {
        return paused;
    }

    /**
     * @dev Mints token.
     *
     * Note: On Rinkeby Testnet MAX_TOKENS has no limit. See the commented out code below.
     */
    function mintToken(uint256 numberTokens)
        public
        payable
        nonReentrant
        returns (uint256[] memory tokenIDs)
    {
        require(!paused, "Sale paused");
        
        require(
            totalSupply() + MAX_RESERVED_TOKENS - reservedTokenCounter <
                MAX_TOKENS + numRemixMashups,
            "Sold out"
        );
        
        require(
            numberTokens > 0 && numberTokens <= 12,
            "Incorrect number of tokens"
        );
        
        require(
            totalSupply() + numberTokens <=
                MAX_TOKENS -
                    MAX_RESERVED_TOKENS +
                    reservedTokenCounter +
                    numRemixMashups,
            "Requested too many tokens"
        );
        
        require(msg.value >= price * numberTokens, "Incorrect amount");
        uint256 mintCost = price * numberTokens;
        uint256 refund = msg.value - mintCost;
        if (refund > 0) {
            Address.sendValue(payable(msg.sender), refund);
        }

        tokenIDs = new uint256[](numberTokens);

        for (uint256 i = 0; i < numberTokens; i++) {
            uint256 mintIndex = tokenCounter++;
            tokenCRNA[mintIndex] =
                mix(cRNA_U.numberLength - 2 * cRNA_U.unitLength, mintIndex) *
                10**(2 * cRNA_U.unitLength);
            tokenIDs[i] = mintIndex;
            _safeMint(_msgSender(), mintIndex);
        }
        return tokenIDs;
    }

    /**
     * @dev Burns a specific ERC721 token.
     * @param tokenId uint256 id of the ERC721 token to be burned.
     */
    function burn(uint256 tokenId) public nonReentrant {
        require(_isApprovedOrOwner(msg.sender, tokenId), "Not the token owner");
        _burn(tokenId);

        if (isRemixed[tokenId] || isMashedUp[tokenId]) {
            numRemixMashups--;
        }

        delete tokenCRNA[tokenId];
        delete isRemixed[tokenId];
        delete remixValue[tokenId];
        delete isMashedUp[tokenId];
        delete mashupValue[tokenId];

        address mashupAddress = mashedUpBy[tokenId];
        uint256 timesMashedUp = mashupRanking[mashupAddress];
        if (timesMashedUp > 0) {
            mashupRanking[mashupAddress] = timesMashedUp - 1;
        }
        delete mashedUpBy[tokenId];
    }

    /**
     * @dev Sets token price.
     *
     * Note: Cannot be changed if sale active.
     */
    function setPrice(uint256 _newPrice) public onlyOwner {
        require(paused, "Price cannot be changed");
        price = _newPrice;
    }

    function withdraw() external onlyOwner {
        uint256 balance = address(this).balance;
        Address.sendValue(payable(owner()), balance);
    }

    /**
     * @dev Mints reserved token.
     */
    function ownerMintToken(uint256 numberTokens)
        public
        nonReentrant
        onlyOwner
    {
        require(reservedTokenCounter < MAX_RESERVED_TOKENS, "Sold out");
        require(
            reservedTokenCounter + numberTokens <= MAX_RESERVED_TOKENS,
            "Requested too many tokens"
        );

        for (uint256 i = 0; i < numberTokens; i++) {
            uint256 mintIndex = tokenCounter++;
            reservedTokenCounter++;
            tokenCRNA[mintIndex] =
                mix(cRNA_U.numberLength - 2 * cRNA_U.unitLength, mintIndex) *
                10**(2 * cRNA_U.unitLength);
            _safeMint(_msgSender(), mintIndex);
        }
    }

    /**
     * @dev See {ICulturalContract-remix}.
     */
    function remix(uint256 tokenId, string memory remixInput)
        public
        override
        nonReentrant
    {
        require(ERC721.ownerOf(tokenId) == _msgSender(), "Not the token owner");
        require(!isRemixed[tokenId], "Remixed token");
        uint256 remixTokenId = tokenCounter++;
        uint256 remixCRNA = mix(cRNA_U.unitLength, tokenId);
        remixValue[remixCRNA] = remixInput;

        tokenCRNA[remixTokenId] = getRemixCRNA(tokenId, remixCRNA);
        if (isMashedUp[tokenId]) {
            isMashedUp[remixTokenId] = true;
        }
        isRemixed[remixTokenId] = true;
        numRemixMashups++;
        _safeMint(_msgSender(), remixTokenId);
        emit Remixed(remixTokenId);
    }

    /**
     * @dev See {ICulturalContract-mashupFee}.
     */
    function mashupFee(address mashupAddress)
        public
        override
        returns (uint256)
    {
        ICulturalContract cContract = ICulturalContract(mashupAddress);
        uint256 mashupCharge = cContract.mashupFee(mashupAddress);
        mashupPrice[mashupAddress] = mashupCharge;
        return mashupCharge;
    }

    /**
     * @dev See {ICulturalContract-mashup}.
     */
    function mashup(uint256 tokenId, address payable mashupAddress)
        public
        payable
        override
        nonReentrant
        returns (string memory)
    {
        require(ERC721.ownerOf(tokenId) == _msgSender(), "Not the token owner");
        require(!isMashedUp[tokenId], "Mashed up token");
        require(msg.value == mashupFee(mashupAddress), "Incorrect amount");
        uint256 mashupTokenId = tokenCounter++;
        uint256 mashupCRNA = mix(cRNA_U.unitLength, tokenId);
        ICulturalContract cContract = ICulturalContract(mashupAddress);
        mashupValue[mashupCRNA] = string(
            abi.encodePacked(cContract.mashup(tokenId, payable(address(this))))
        );
        tokenCRNA[mashupTokenId] = getMashupCRNA(tokenId, mashupCRNA);
        if (isRemixed[tokenId]) {
            isRemixed[mashupTokenId] = true;
        }
        isMashedUp[mashupTokenId] = true;
        mashedUpBy[mashupTokenId] = mashupAddress;
        numRemixMashups++;
        _safeMint(_msgSender(), mashupTokenId);
        Address.sendValue(payable(mashupAddress), msg.value);
        addToRegistry(mashupAddress);
        latestMashupTokenId[mashupAddress] = mashupTokenId;
        ICulturalContractMetadata ccMetadata = ICulturalContractMetadata(
            mashupAddress
        );
        mashupName[mashupAddress] = ccMetadata.ccName();
        mashupSymbol[mashupAddress] = ccMetadata.ccSymbol();
        mashupAuthor[mashupAddress] = ccMetadata.ccAuthor();
        emit MashedUp(mashupTokenId);
        return "";
    }

    /**
     * @dev Provides preview for a given mashup
     */
    function mashupURI(address mashupAddress)
        public
        view
        returns (string memory)
    {
        uint256 tokenId = 0;
        tokenId = latestMashupTokenId[mashupAddress];

        string
            memory svgFilePrev = '<svg xmlns="http://www.w3.org/2000/svg" width="100%" height="100%" preserveAspectRatio="xMinYMin meet" viewBox="0 0 350 350"><style type="text/css">.txt {fill: white;font-family: "Open Sans", sans-serif;font-size: 3em;}</style><rect width="100%" height="100%" fill="lightgray" /><text x="90" y="150" class="txt">Mashup</text><text x="90" y="210" class="txt">Preview</text></svg>';
        string[7] memory svgString;

        svgString[
            0
        ] = '<svg xmlns="http://www.w3.org/2000/svg" width="100%" height="100%" preserveAspectRatio="xMinYMin meet" viewBox="0 0 350 350"><style type="text/css">.txt{fill: black;font-family: "Open Sans", sans-serif; font-size: 0.75em;}</style><rect width="100%" height="100%" fill="';
        svgString[1] = getColor(tokenId);
        svgString[2] = '"/>';
        svgString[3] = getMashedUp(tokenId);
        svgString[4] = '<text x="5" y="310" class="txt">';
        svgString[5] = getText(tokenId);
        svgString[6] = "</text></svg>";

        string memory svgFile = string(
            abi.encodePacked(
                svgString[0],
                svgString[1],
                svgString[2],
                svgString[3],
                svgString[4],
                svgString[5],
                svgString[6]
            )
        );
        if (tokenId > 0 && _exists(tokenId)) {
            return string(abi.encodePacked(Base64.encode(bytes(svgFile))));
        } else {
            return string(abi.encodePacked(Base64.encode(bytes(svgFilePrev))));
        }
    }

    function getBalance() public view returns (uint256) {
        return address(this).balance;
    }

    /**
     * @dev See {ICulturalContractMetadata-ccName}.
     */
    function ccName() public view virtual override returns (string memory) {
        return ERC721.name();
    }

    /**
     * @dev See {ICulturalContractMetadata-ccSymbol}.
     */
    function ccSymbol() public view virtual override returns (string memory) {
        return ERC721.symbol();
    }

    /**
     * @dev See {ICulturalContractMetadata-ccAuthor}.
     */
    function ccAuthor() public view virtual override returns (string memory) {
        return author;
    }

    /**
     * @dev Returns true if token has been remixed.
     */
    function getIsRemixed(uint256 tokenId) public view returns (bool) {
        return isRemixed[tokenId];
    }

    /**
     * @dev Returns true if token has been mashed up.
     */
    function getIsMashedUp(uint256 tokenId) public view returns (bool) {
        return isMashedUp[tokenId];
    }

    /**
     * @dev Returns number of mashups in registry.
     */
    function numMashupsInRegistry() public view returns (uint256) {
        return allMashups.length;
    }

    /**
     * @dev Returns number of mashups plus remixes.
     */
    function totalRemixMashups() public view returns (uint256) {
        return numRemixMashups;
    }

    /**
     * @dev Returns number of minted reserved tokens.
     */
    function reservedTokensMinted() public view returns (uint256) {
        return reservedTokenCounter;
    }

    /**
     * @dev Returns mashup address.
     */
    function getMashupAddress(uint256 atIndex) public view returns (address) {
        return allMashups[atIndex];
    }

    /**
     * @dev Returns number of mashups created by `mashupAddress`.
     */
    function getMashupRanking(address mashupAddress)
        public
        view
        returns (uint256)
    {
        return mashupRanking[mashupAddress];
    }

    /**
     * @dev Returns mashup price for a given `mashupAddress`.
     */
    function getMashupFee(address mashupAddress) public view returns (uint256) {
        return mashupPrice[mashupAddress];
    }

    /**
     * @dev Returns mashup name for a given `mashupAddress`.
     */
    function getMashupName(address mashupAddress)
        public
        view
        returns (string memory)
    {
        return mashupName[mashupAddress];
    }

    /**
     * @dev Returns mashup symbol for a given `mashupAddress`.
     */
    function getMashupSymbol(address mashupAddress)
        public
        view
        returns (string memory)
    {
        return mashupSymbol[mashupAddress];
    }

    /**
     * @dev Returns mashup author for a given `mashupAddress`.
     */
    function getMashupAuthor(address mashupAddress)
        public
        view
        returns (string memory)
    {
        return mashupAuthor[mashupAddress];
    }

    /**
     * @dev Adds `mashupAddress` to Mashup Registry.
     */
    function addToRegistry(address mashupAddress) internal {
        if (!inMashupRegistry[mashupAddress]) {
            allMashups.push(mashupAddress);
            allMashupsIndex[mashupAddress] = allMashups.length - 1;
            inMashupRegistry[mashupAddress] = true;
            mashupRanking[mashupAddress] = 1;
        } else {
            uint256 timesMinted = mashupRanking[mashupAddress];
            mashupRanking[mashupAddress] = timesMinted + 1;
        }
    }

    /**
     * @dev Removes `mashupAddress` from Mashup Registry.
     */
    function removeFromRegistry(address mashupAddress) external onlyOwner {
        if (inMashupRegistry[mashupAddress]) {
            uint256 lastMashupIndex = allMashups.length - 1;
            uint256 mashupIndex = allMashupsIndex[mashupAddress];
            address lastMashupAddress = allMashups[lastMashupIndex];
            allMashups[mashupIndex] = lastMashupAddress;
            allMashupsIndex[lastMashupAddress] = mashupIndex;
            delete allMashupsIndex[mashupAddress];
            allMashups.pop();
            inMashupRegistry[mashupAddress] = false;
        }
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/ERC721Enumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/ERC721Enumerable.sol)

pragma solidity ^0.8.0;

import "../ERC721.sol";
import "./IERC721Enumerable.sol";

/**
 * @dev This implements an optional extension of {ERC721} defined in the EIP that adds
 * enumerability of all the token ids in the contract as well as all token ids owned by each
 * account.
 */
abstract contract ERC721Enumerable is ERC721, IERC721Enumerable {
    // Mapping from owner to list of owned token IDs
    mapping(address => mapping(uint256 => uint256)) private _ownedTokens;

    // Mapping from token ID to index of the owner tokens list
    mapping(uint256 => uint256) private _ownedTokensIndex;

    // Array with all token ids, used for enumeration
    uint256[] private _allTokens;

    // Mapping from token id to position in the allTokens array
    mapping(uint256 => uint256) private _allTokensIndex;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC721) returns (bool) {
        return interfaceId == type(IERC721Enumerable).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721Enumerable-tokenOfOwnerByIndex}.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721.balanceOf(owner), "ERC721Enumerable: owner index out of bounds");
        return _ownedTokens[owner][index];
    }

    /**
     * @dev See {IERC721Enumerable-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _allTokens.length;
    }

    /**
     * @dev See {IERC721Enumerable-tokenByIndex}.
     */
    function tokenByIndex(uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721Enumerable.totalSupply(), "ERC721Enumerable: global index out of bounds");
        return _allTokens[index];
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual override {
        super._beforeTokenTransfer(from, to, tokenId);

        if (from == address(0)) {
            _addTokenToAllTokensEnumeration(tokenId);
        } else if (from != to) {
            _removeTokenFromOwnerEnumeration(from, tokenId);
        }
        if (to == address(0)) {
            _removeTokenFromAllTokensEnumeration(tokenId);
        } else if (to != from) {
            _addTokenToOwnerEnumeration(to, tokenId);
        }
    }

    /**
     * @dev Private function to add a token to this extension's ownership-tracking data structures.
     * @param to address representing the new owner of the given token ID
     * @param tokenId uint256 ID of the token to be added to the tokens list of the given address
     */
    function _addTokenToOwnerEnumeration(address to, uint256 tokenId) private {
        uint256 length = ERC721.balanceOf(to);
        _ownedTokens[to][length] = tokenId;
        _ownedTokensIndex[tokenId] = length;
    }

    /**
     * @dev Private function to add a token to this extension's token tracking data structures.
     * @param tokenId uint256 ID of the token to be added to the tokens list
     */
    function _addTokenToAllTokensEnumeration(uint256 tokenId) private {
        _allTokensIndex[tokenId] = _allTokens.length;
        _allTokens.push(tokenId);
    }

    /**
     * @dev Private function to remove a token from this extension's ownership-tracking data structures. Note that
     * while the token is not assigned a new owner, the `_ownedTokensIndex` mapping is _not_ updated: this allows for
     * gas optimizations e.g. when performing a transfer operation (avoiding double writes).
     * This has O(1) time complexity, but alters the order of the _ownedTokens array.
     * @param from address representing the previous owner of the given token ID
     * @param tokenId uint256 ID of the token to be removed from the tokens list of the given address
     */
    function _removeTokenFromOwnerEnumeration(address from, uint256 tokenId) private {
        // To prevent a gap in from's tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = ERC721.balanceOf(from) - 1;
        uint256 tokenIndex = _ownedTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary
        if (tokenIndex != lastTokenIndex) {
            uint256 lastTokenId = _ownedTokens[from][lastTokenIndex];

            _ownedTokens[from][tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
            _ownedTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index
        }

        // This also deletes the contents at the last position of the array
        delete _ownedTokensIndex[tokenId];
        delete _ownedTokens[from][lastTokenIndex];
    }

    /**
     * @dev Private function to remove a token from this extension's token tracking data structures.
     * This has O(1) time complexity, but alters the order of the _allTokens array.
     * @param tokenId uint256 ID of the token to be removed from the tokens list
     */
    function _removeTokenFromAllTokensEnumeration(uint256 tokenId) private {
        // To prevent a gap in the tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = _allTokens.length - 1;
        uint256 tokenIndex = _allTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary. However, since this occurs so
        // rarely (when the last minted token is burnt) that we still do the swap here to avoid the gas cost of adding
        // an 'if' statement (like in _removeTokenFromOwnerEnumeration)
        uint256 lastTokenId = _allTokens[lastTokenIndex];

        _allTokens[tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
        _allTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index

        // This also deletes the contents at the last position of the array
        delete _allTokensIndex[tokenId];
        _allTokens.pop();
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: base64-sol/base64.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0;

/// @title Base64
/// @author Brecht Devos - <brecht@loopring.org>
/// @notice Provides functions for encoding/decoding base64
library Base64 {
    string internal constant TABLE_ENCODE = 'ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/';
    bytes  internal constant TABLE_DECODE = hex"0000000000000000000000000000000000000000000000000000000000000000"
                                            hex"00000000000000000000003e0000003f3435363738393a3b3c3d000000000000"
                                            hex"00000102030405060708090a0b0c0d0e0f101112131415161718190000000000"
                                            hex"001a1b1c1d1e1f202122232425262728292a2b2c2d2e2f303132330000000000";

    function encode(bytes memory data) internal pure returns (string memory) {
        if (data.length == 0) return '';

        // load the table into memory
        string memory table = TABLE_ENCODE;

        // multiply by 4/3 rounded up
        uint256 encodedLen = 4 * ((data.length + 2) / 3);

        // add some extra buffer at the end required for the writing
        string memory result = new string(encodedLen + 32);

        assembly {
            // set the actual output length
            mstore(result, encodedLen)

            // prepare the lookup table
            let tablePtr := add(table, 1)

            // input ptr
            let dataPtr := data
            let endPtr := add(dataPtr, mload(data))

            // result ptr, jump over length
            let resultPtr := add(result, 32)

            // run over the input, 3 bytes at a time
            for {} lt(dataPtr, endPtr) {}
            {
                // read 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // write 4 characters
                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1)
                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1)
                mstore8(resultPtr, mload(add(tablePtr, and(shr( 6, input), 0x3F))))
                resultPtr := add(resultPtr, 1)
                mstore8(resultPtr, mload(add(tablePtr, and(        input,  0x3F))))
                resultPtr := add(resultPtr, 1)
            }

            // padding with '='
            switch mod(mload(data), 3)
            case 1 { mstore(sub(resultPtr, 2), shl(240, 0x3d3d)) }
            case 2 { mstore(sub(resultPtr, 1), shl(248, 0x3d)) }
        }

        return result;
    }

    function decode(string memory _data) internal pure returns (bytes memory) {
        bytes memory data = bytes(_data);

        if (data.length == 0) return new bytes(0);
        require(data.length % 4 == 0, "invalid base64 decoder input");

        // load the table into memory
        bytes memory table = TABLE_DECODE;

        // every 4 characters represent 3 bytes
        uint256 decodedLen = (data.length / 4) * 3;

        // add some extra buffer at the end required for the writing
        bytes memory result = new bytes(decodedLen + 32);

        assembly {
            // padding with '='
            let lastBytes := mload(add(data, mload(data)))
            if eq(and(lastBytes, 0xFF), 0x3d) {
                decodedLen := sub(decodedLen, 1)
                if eq(and(lastBytes, 0xFFFF), 0x3d3d) {
                    decodedLen := sub(decodedLen, 1)
                }
            }

            // set the actual output length
            mstore(result, decodedLen)

            // prepare the lookup table
            let tablePtr := add(table, 1)

            // input ptr
            let dataPtr := data
            let endPtr := add(dataPtr, mload(data))

            // result ptr, jump over length
            let resultPtr := add(result, 32)

            // run over the input, 4 characters at a time
            for {} lt(dataPtr, endPtr) {}
            {
               // read 4 characters
               dataPtr := add(dataPtr, 4)
               let input := mload(dataPtr)

               // write 3 bytes
               let output := add(
                   add(
                       shl(18, and(mload(add(tablePtr, and(shr(24, input), 0xFF))), 0xFF)),
                       shl(12, and(mload(add(tablePtr, and(shr(16, input), 0xFF))), 0xFF))),
                   add(
                       shl( 6, and(mload(add(tablePtr, and(shr( 8, input), 0xFF))), 0xFF)),
                               and(mload(add(tablePtr, and(        input , 0xFF))), 0xFF)
                    )
                )
                mstore(resultPtr, shl(232, output))
                resultPtr := add(resultPtr, 3)
            }
        }

        return result;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits a {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Enumerable.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {
    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256 tokenId);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}